MOTOROLA SEMICONDUCTOR APPLICATION NOTE

Using M68HC16 Digital Signal Processing To Build An Audio Frequency Analyzer

By Mark Glenewinkel

INTRODUCTION

This application note demonstrates the use of a microcontroller unit (MCU) with integrated DSP capabilities. The MC68HC16Z1 is a high performance 16-bit MCU that includes on-chip peripheral modules and a CPU module (CPU16). The CPU16 instruction set simplifies the use of digital signal processing algorithms, and makes it easy to implement low-bandwidth filter and control-oriented applications.

OBJECTIVES

The goal of this application note is for an engineer to learn the MC68HC16Z1 well enough to design and build an audio frequency analyzer (AFA). The following intermediate objectives have been defined to help reach this goal.

- Learning the CPU16 instruction set
- Becoming familiar with MC68HC16Z1 modules
- Learning basic MCU I/O hardware and software
- Understanding DSP system concepts with the frequency analyzer
- Understanding and implementing common DSP algorithms with an MCU

This is a tutorial design project that follows a hands-on approach to using DSP. It provides concrete hardware/software applications that are used to understand and design an MCU-based system utilizing DSP algorithms. A basic knowledge of MC68HC16Z1 hardware and the CPU16 instruction set is necessary to complete the design project. A complete discussion of digital signal processing is beyond the scope of this note. However, there are a number of standard textbooks and references available. Please refer to the Motorola publications listed under **REFERENCES** for more information concerning topics and devices discussed in this note.



EQUIPMENT REQUIRED

The following items are needed to build and test the audio frequency analyzer (AFA).

- 1. An IBM PC compatible computer with a parallel printer port
- 2. The M68HC16Z1EVB
- 3. A prototyping or wire-wrap board
- 4. One straight DB25 cable, male on one end, female on the other
- 5. A 5 volt power supply
- 6. An audio sound source, preferably a CD player
- 7. Two Y-connectors to split the stereo sound source with audio cables
- 8. A sinusoidal waveform generator, optional
- 9. Oscilloscope for debugging, optional

All of the components needed to build the AFA are shown in Figure 4 and Figure 5, the AFA schematics.

THE AUDIO FREQUENCY ANALYZER

Spectral analysis is a method of determining the specific frequency content of a signal and the energy levels of these frequencies. This information is processed by either Fourier Transform methods or by specific filtering of the signal. The information is tabulated for more analysis or displayed in a visual format.

One example of spectral analysis is found in oil exploration. An engineer sends a known signal into the earth and then calculates the frequency content of the reflected signal. This is a classic input/output black box. The transfer function of the black box (the earth in this case) yields clues to the structure beneath the surface. Different frequency responses correspond to different types of rock. With spectral analysis, the engineer can decide whether it is feasible to drill.

This project focuses on the frequency analysis of an audio signal. A frequency analyzer is often used in audio systems and recording studios. It filters out energy levels of specific audio frequencies and displays them to indicate the frequency content of the audio signal. Audio frequency analyzers are also used in conjunction with equalizers to help the user define and shape the spectral characteristics of a sound source.

Figure 1 is a generic system diagram of a frequency analyzer based on bandpass filters. The input signal is split and sent to all the filters. The filters pass only specific frequency components of the input signal. After filtration, the strength of each passed signal is analyzed, and the amount of energy in each band is represented on an LED display. This process is executed in a continuous real-time algorithm. **Figure 2** shows a typical audio frequency analyzer transfer function.

Figure 3 is a system diagram of the AFA project, which is implemented using digital filters. Two stereo audio signal inputs are combined by a summing circuit. An anti-aliasing filter removes unwanted high frequency components. A biasing circuit centers the signal around 2.5 vdc for proper analog-to-digital conversion. The ADC module in the MC68HC16Z1 samples the analog signal and digitizes it, then the data is processed by the CPU16. Processing consists of five DSP bandpass filter algorithms. Each determines the amplitude of a specific frequency band and encodes display data. The queued serial peripheral interface (QSPI) is used to send display data to the LED array in real time. Each of these functional blocks is discussed in detail later in this note. Hardware is discussed first, then software.



Figure 1 Frequency Analyzer System Diagram



Figure 2 Bandpass Frequency Analyzer Transfer Function



AN1233 F3

Figure 3 Audio Frequency Analyzer System Diagram

AFA Hardware

Familiarity with the AFA hardware helps to understand the code used to implement the analyzer. **Figure 4** is a schematic of the analog front end of the AFA, and **Figure 5** is a schematic of the display logic.

Figure 4 AFA Analog Front End

The Analog Front End

The analog front end contains all of the circuitry to condition the signal for analog-to-digital conversion and subsequent digital signal processing. It consists of the summing circuitry for the stereo signal, the anti-aliasing filter, and the biasing circuitry for the ADC. A MAX274 low-pass filter chip, manufactured by the Maxim Corporation of Sunnyvale, California, is used to implement all of these functions.

The MAX274 is an eighth order, programmable, continuous-time active filter. The chip consists of four independent cascadable second-order filter sections. Each filter section can implement any all-pole bandpass or lowpass filter, characterized as a Butterworth, Bessel, or Chebyshev response. Each second-order section is programmable with four external resistors. A second-order section is illustrated in **Figure 6**. Maxim provides an evaluation board and a software package that calculates resistor values from response specifications input by the user. This makes the MAX274 very flexible and easy to use when implementing highorder anti-aliasing filters.

The Summing Amplifier

The summing amplifier combines the two analog stereo signals coming into the system from the audio source. The basic summing circuit shown in **Figure 3** is implemented in the AFA by using an op amp in the first second-order filter section of the MAX274. As shown in **Figure 4**, two summing resistors (R14 and R16) are used to feed the input signals to the inverting input of the op amp, which combines them into one signal.

Anti-Aliasing Filter

When a signal of a given frequency is sampled at too low a rate, it appears as a totally different lower frequency at the output of the sampler. This phenomenon is referred to as aliasing. Aliasing occurs at a point called the folding frequency, which is one-half the sampling frequency. In order for the frequency analyzer to be accurate, sampling frequency must therefore be at least two times the highest frequency component to be sampled. The ideal solution to this problem is to raise the sampling rate as high as possible, but realworld designs generally have a fixed upper limit on sampling frequency. The most practical solution is to attenuate high frequency components of the input signal so that aliasing does not occur. The anti-aliasing filter correctly attenuates the high frequency components of the signal, so that they are not present within the sample bandwidth.

The AFA has a 25-kHz sampling frequency (Fs), and a processing bandwidth of 10 kHz. If no filter is used, signal components with a frequency higher than 12.5 kHz alias at lower frequencies, and the digitized samples represent invalid information. **Figure 7** shows these relationships. Fs/2 is the folding frequency, 12.5 kHz. Frequencies that will not alias with a 25 kHz sampling frequency are to the left of Fs/2, while frequencies that will alias are to the right of Fs/2.

Figure 7 AFA Aliasing Without Filter

Anti-aliasing filter design must be a compromise. An efficient and economical solution is to find an intermediate filtration range, between high-order filter roll-off and DSP bandwidth. If the filter has a slow roll-off, a higher sampling frequency is needed, the sampling period is shortened, and there is less time for the DSP algorithm to execute. In other words, a steeper roll-off requires a lower sampling frequency, which in turn provides a longer sampling period for DSP operation.

The AFA anti-aliasing filter passes frequencies up to 10 kHz. The filter stop band begins at 15 kHz. Stop band attenuation is dependent upon the application and the dynamic range of the sampled data. In this case, filter output is fed to the ADC module in the MC68HC16Z1, which does not have sufficient resolution or dynamic range to "see" energy in the stop band. An 8-bit conversion that allows a dynamic range of 48 dB is used. The following equation shows these relationships.

Where:

ADCres = A/D converter resolution

System bandwidth is 10 kHz, and at a 25 kHz sampling frequency, components above 12.5 kHz will alias. Therefore, the signal must be attenuated 48 dB to eliminate all aliasing components. Accordingly, the filter must have a minimum drop-off slope of 96 dB per octave. To insure that this requirement is met, a roll-off of 100 dB per octave is used. Using these values with the MAX274 design software, resistor values for an eighth order 0.5 dB passband ripple Chebyshev filter were obtained. Lower passband ripple was sacrificed to gain steeper roll-off. The anti-aliasing filter response programmed into the MAX274 is shown in **Figure 8**.

Figure 8 AFA Anti-Aliasing Filter Roll-Off

ADC Input Biasing

The MC68HC16Z1 ADC module can convert analog data into six different digital representations. Digital data can have 8-bit or 10-bit resolution, can be signed or unsigned, and can be left or right justified. These formats are shown in **Figure 9**.

AN1233 F9

Figure 9 ADC Conversion Formats

Figure 10 shows hexadecimal representations of signed and unsigned ADC data. For 8-bit conversions, there are 256 possible values. Unsigned formats assume the zero voltage point is at the low ADC reference voltage, with 256 steps from low to high reference. Signed formats assume that the zero voltage point is halfway between the low and high ADC reference voltages. The most significant bit indicates a positive or negative value — 128 values represent positive voltages, and 128 two's-complement values represent negative voltages (\$00 represents the midpoint, and \$FF represents midpoint minus one count).

AN1233 F10

Figure 10 Hexadecimal Representation of 8-Bit ADC Data

The AFA uses signed, 8-bit, left justified ADC data. The analog signal must be biased at 2.5 vdc, centered between the 0 vdc and 5 vdc ADC reference voltages, in order to use this representation. The MAX274 is used to bias the signal.

The MAX274 requires two power connections. Biasing circuitry consists of a voltage divider (R12, R13) and decoupling capacitors (C10 – C13) connected to one of the MAX274 supplies. The V– pin is connected to analog ground. The V+ pin is connected to the 5 volt supply. The GND pin is connected to 2.5 volts. This splits the supply and causes the analog signal to have a 2.5 volt DC offset. The signal is buffered by an op amp driver and is sent directly to the ADC module pins from the MAX274. The ADC can now properly sample the signal.

The Digital Back End

The digital back end shown in **Figure 5** contains all of the circuitry required to output digitally processed information to the LED array. When digital signal processing is complete, encoded energy levels for each band are loaded into QSPI transmit RAM, then the QSPI is activated, and the data is transmitted serially to the MC14489 LED drivers.

QSPI software is one of the more difficult aspects of the AFA, but the hardware is quite simple. Three QSPI pins, MOSI, SCK, and PCS0, are used. The master out slave in (MOSI) connection is used to transfer data, the serial clock (SCK) connection is used to clock the transfer, and the peripheral chip select (PCS0) connection is used to enable the LED drivers. The QSPI must be configured correctly to transfer data to the drivers. Refer to the *QSM Reference Manual* (QSMRM/AD) for more information about the QSPI.

The MC14489 LED Driver

The MC14489 can drive individual lamps, seven-segment displays, or combinations of both, in a multiplexed fashion. The chip receives data via a serial input port, and features data retention plus decode and scan circuitry. This reduces software overhead required to perform these tasks. A single current-limiting resistor (Rx) is the only external component needed to operate the MC14489.

Three MC14489 drivers are used in the AFA. There are five 8-bit LED arrays. Two of the MC14489 chips control four banks of four diodes each, and one controls two banks of four diodes each. Drive current for diodes in each bank is supplied by pins A, B, C, and D of the MC14489. The cathodes of each bank of diodes are tied together and a bank-select pin sinks the current for that bank. Please refer to the MC14489 Data Sheet for more information.

The M68HC16Z1 EVB and Development Environment

The M68HC16Z1 Evaluation Board provides the capability to test and debug the audio frequency analyzer. **Table 1** shows development software supplied with the EVB.

MASM16.EXE	
MASM.EXE	
HEX.EXE	
MASM16.HLP	
EVB16.EXE	

Table 1 Development Software

MASM16 software is used to edit and assemble code, and EVB16 software is used to download code to the EVB and run it. EVB16 software also has debug capabilities such as trace and breakpoint. Please refer to the *M68HC16Z1EVB User's Manual* for a list of debug features.

Assembling the Development Environment

Assembling the development system with the AFA is simple. Hook up the system as shown in **Figure 11**. The AFA project board connects to the M68HC16Z1EVB via P7 and P6. Use the DB25 cable to connect the parallel port of the PC to the parallel port connector of the EVB. After connecting the 5 volt power supply to the M68HC16Z1EVB, connect the audio signal source. A CD player is the recommended source for a high quality output. Split the audio source outputs so that both the AFA board and the speakers receive the signals (audio splitters can be found at most stereo and electronics stores).

Figure 11 AFA Development System Setup

AFA Software

Even though hardware is required to build the AFA, software running on the CPU16 performs most of the actual work. Five tutorial programs must be integrated to complete the project. Each program demonstrates specific functions of the AFA, and each is discussed in a separate section. Since this is a DSP project/tutorial, discussion focuses on signal-processing tasks. Each of the tutorial programs must be modified in order to complete the AFA. The software steps to the AFA design are listed below.

- 1. Acquisition of data
- 2. QSPI to MC14489 interface
- 3. Periodic interrupt timer routine
- 4. Peak detector
- 5. 1-kHz bandpass filter routine
- 6. 5-band audio frequency analyzer

AFA software is listed in **Table 2**. Each of the first six programs in the table corresponds to one of the software steps listed above. In order to organize and streamline the project, each program has been designed according to a standard template for the M68HC16Z1EVB. **Figure 12** shows the template.

ADC.ASM
QSPI_LED.ASM
INT_TEST.ASM
PEAK.ASM
1K_FLTR.ASM
5BAND_SA.ASM
EQUATES.ASM
ORG00000.ASM
INITSYS.ASM
INITRAM.ASM
OUTVAL1.ASM
OUTVAL2.ASM

Table 2 AFA Project Software

OUTVAL1.ASM and OUTVAL2.ASM are lookup tables for the LED display routines. They contain values that correspond to the number of LEDs needed to reflect a given peak value.

In addition, utility files that simplify startup and usage of the MC68HC16Z1 have been included in the AFA software package. A brief description of each include file follows.

EQUATES.ASM provides an equates table of MC68HC16Z1 registers and equivalent address values.

ORG00000.ASM defines the reset vector.

INITSYS.ASM initializes the CPU16, takes care of the extension registers, disables the COP watchdog, and sets system clock speed to 16.78 MHz.

INITRAM.ASM turns on the 1-Kbyte SRAM module, maps the RAM array to address \$10000, and moves the stack pointer to \$103FE to increase interrupt-processing speed.

Source code for all of these files is available on the Motorola Freeware Bulletin Board. The BBS number is (512) 891-3733. The files are archived under the name AFA.ARC, in the AMCU section.

* MOTOROLA, INC. Advanced MCU Division * * Austin, Texas * Title: HC16 SOFTWARE TEMPLATE File Name: TEMPLATE.ASM Description: This program provides a template for all designers to use with the HC16Z1 An equate table is given. * The reset vector is initialized. * The CPU and RAM are also initialized. The user can put his code in the 'user area' * block of this template * History: 06/05/91 Created. * 10/02/91 Modified comments. * Note: This program is written for the M68HC16Z1EVB. * * * * * * * * * * * * * * * INCLUDE 'EQUATES.ASM' ;table of EQUates for common register addr INCLUDE 'ORG00000.ASM' ; initialize reset vector ORG \$0200 ;start program after interrupt vectors Initialization Routines ***** **** INCLUDE 'INITSYS.ASM' ; initially set EK=F, XK=0, YK=0, ZK=0 ;set sys clock at 16.78 MHz, disable COP INCLUDE 'INITRAM.ASM' ; initialize and turn on SRAM ;set stack (SK=1, SP=03FE) ***** Start of user program area *****

Figure 12 AFA Software Template

Software Design Constraints

It is important to understand the specifications and system constraints on the software. A software flow diagram of the AFA is shown in **Figure 13**. Each of the process boxes in the flowchart corresponds to one of the steps toward the complete design. The main tasks are to convert analog input to digital data, run five infinite impulse response (IIR) bandpass filter routines, detect the peak amplitude of each filtered signal, encode the peak value to an LED display value, update the QSPI transmit RAM, and transmit the information to the LED drivers. The flowchart also shows that the AFA is a real-time digital signal processing algorithm that runs in a continuous loop.

Figure 13 AFA System Software Flowchart

All processing must be completed within one period of the 24.95-kHz sampling frequency. As shown below, a 24.95-kHz sampling frequency is equivalent to a 40.08-µs sampling period. The MC68HC16Z1 is running at 16.78 MHz, so the system clock period is 60 ns. Thus, all necessary processing must be completed in 668 clock cycles, before the next sampling period begins.

Fs = 24.95 kHz Ts = 1/Fs = 40.08 μs Fc = 16.78 MHz Tc = 1/Fc = 60 ns

System clock cycles per sampling period = Ts/Tc = 668 system clock cycles

Where:

- Fs = Sampling frequency
- Ts = Sampling period
- Fc = MC68HC16Z1 CPU clock frequency
- Tc = MC68HC16Z1 CPU clock period

AN1233 F13

Figure 14 shows the relationship between sampling periods and real-time digital signal processing. All calculations and internal/external housekeeping must be taken care of within the given sample period.

AN1233 F14

Figure 14 AFA Sampling Period

Software Design Implementation

The following sections examine AFA software in detail. For each of the programs, there is a discussion of design and implementation, a code listing, and appropriate flow charts. In the interest of brevity, the standard template headers have been omitted from the listings, and redundant portions of flowcharts are reproduced only once.

Analog-to-Digital Data Acquisition (ADC.ASM)

In order to perform digital signal processing, a digital representation of the analog signal must be available. The MC68HC16Z1 contains a programmable ADC module. The ADC has a number of automatic conversion modes. Only four registers are needed to control the ADC. Refer to the *ADC Reference Manual* (ADCRM/AD) for more detailed information.

ADC.ASM initializes the ADC module, then goes into a continuous loop, repeating the programmed conversion sequence. **Figure 15** is a flowchart of **ADC.ASM**.

To test the routine, first load and assemble the **ADC.ASM** file, then switch to the EVB16 debugger. Download the assembled file to the M68HC16Z1EVB, trace execution until the infinite loop begins to execute, then examine the ADC result registers.

Display the memory locations starting at \$FF710. Check the memory location \$FF711. If the AFA is hooked up properly, a value somewhere between \$74 and \$8B will be displayed. This value is an unsigned representation of 2.5 volts, plus or minus the offset voltage of the MAX274. This same value should also be found at location \$FF730. The signed representation of the same data is found at location \$FF720. The design of **ADC.ASM** is finished. Some of this code will be used to build other programs.

ADC.ASM Code listing

	INCLUDE INCLUDE	'EQUATES.ASM' 'ORG00000.ASM'	;table of EQUates for common register addr ;initialize reset vector
	ORG	\$0200	
* * * * *	Initiali	zation Routines	****
	INCLUDE	'INITSYS.ASM'	; initially set EK=F, XK=0, YK=0, ZK=0
	INCLUDE	'INITRAM.ASM'	;set sys clock at 16.78 MHz, disable COP ;initialize and turn on SRAM ;set stack (SK=1, SP=03FE)
ORG ****	\$0200 ADC Ini	tialization	****
	STD	ADCMCR	;turn on ADC
	STD	ADCTL0	;8-bit, set sample period
****	ADC Sta	rt	****
LOOP	LDD STD	#\$0000 ADCTL1	;single 4 conversion, single channel, AD0 ;writing to the ADCTL1 reg starts conversion
SCFSET	LDAA BITA BEQ BRA	#\$80 ADSTAT SCFSET LOOP	;check for the Sequence Complete Flag ;complete?, if not check again ;go get another sample
		START	
		INCLUDE 'EQUATES	3. ASM THESE INCLUDE FILES CONTAIN THE FOLLOWING: AN EQUATE FILE WITH ALL THE Z1 REGISTERS DEFINED

AN1233 F15

Figure 15 ADC.ASM Flowchart

QSPI TO MC14489 Interface (QSPI_LED.ASM)

This program illustrates QSPI serial timing and data format, which must be understood in order to program the QSPI to talk to the MC14489. The *QSM Reference Manual* (QSMRM/AD) and the MC14489 data sheet are needed to understand the code.

QSPI_LED.ASM initializes the QSPI module and the three MC14489 drivers to handle 40 LEDs. After this it updates the LED array by writing to the MC14489 display registers, then gives control back to the EVB16 development software. Values being sent to the array may be changed either by modifying the memory locations that hold the transmitted data or by reassembling the lines that load these memory locations. Figure 16 is a flowchart of **QSPI_LED.ASM**.

QSPI_LED.ASM Code Listing

	INCLUDE INCLUDE	'EQUATES.ASM' 'ORG00000.ASM'	<pre>;table of EQUates for common register addr ;initialize reset vector</pre>
	ORG	\$0200	
* * * * *	Initiali	zation Routines	****
	INCLUDE	'INITSYS.ASM'	; initially set EK=F, XK=0, YK=0, ZK=0 ; set sys clock at 16.78 MHz, disable COP
	INCLUDE	'INITRAM.ASM'	;initialize and turn on SRAM ;set stack (SK=1, SP=03FE)
* * * * *	QSPI In	itialization	* * * *
	STAA	QPDR #\$0F	;output pcs0/ss* to 0 when asserted
	STAA	490F QPAR #CEF	;assign QSM port pins to qspi module
	STAA	QDDR	;assign all QSM pins as outputs except miso
	LDD	#\$8004	;mstr, womq=cpol=cpha=0
	LDD	#\$0300	;16 bits, 2.10MHz serial baud rate ;no interrupt generated, no wrap mode
	STD	SPCR2	;newqp=0, endqp=3, queued for 4 trans
* * * * *	Fill QSI	PI Command.ram t #\$C0	o write the config registers of the 14489
	STAA	CR0	<pre>;cont=1, bitse=1, pcs0=0, no delays needed</pre>
	STAA	CR1 CR2	
	LDAA STAA	#\$40 CR3	;cont=0, bitse=1, pcs0=0, no delays needed
* * * * *	Fill QS	PI Transmit.ram	to write the config registers of the 14489
	LDAA STD	#\$3F TR0+1	;store \$3F to tran.ram registers
	STD STD	TR2 TR3+1	
* * * * *	Turn on	the QSPI, this	will write to the config registers
**** GO	of the l LDD	MC14489 drivers #\$8404	
ODINE	STAA	SPCR1	;turn on spi
SPIWI.	LDAA ANDA	#\$80	spif bit is set, before we can send more
	CMPA BNE	#\$80 SPIWT	;check for spi done
* * * * *	Fill QS	PI Command.ram t	o write the display registers of the 14489
	LDAA STAA	#\$C0 CR0	;cont=1, bitse=1, pcs0=0, no delays needed
	STAA	CR1	
	LDAA STAA	#\$40 CR2	;cont=0, bitse=1, pcs0=0, no delays needed
	STAA	CR4	ant-1 bitan-0 page of the deleter and
	STAA	#\$00 CR3	, cont=1, bitse=0, pcs0=0, no delays needed

****	Fill QSH The beg: LDD STD STAA LDD STD CLRD STD STD STD	PI Transmit.ram f inning LED values #\$8000 TR0 TR3+1 #\$0080 TR1 TR2 TR4	<pre>For display registers of the 14489 s will be \$00, all of the LEDs will be off ;TR0 = \$8000 ;TR1 = \$0080 ;TR2 = \$0000 ;TR3 = \$XX80 ;TR4 = \$0000</pre>
	LDD STD	#\$0400 SPCR2	;display registers need 5 transmissions ;newqp=0, endqp=4
**** TT125	Load up	the various LED	bands for experimentation
1125	STAA	TR4+1	;125 Hz band
т500	LDAA	#\$3F	
	STAA	TR4	;500 Hz band
TIK	LDAA	#ŞFF	the He hand
т4к	SIAA LDAA	1R2+1 #\$3F	, IK HZ DAIIO
	STAA	TR2	;4k Hz band
T10K	LDAA	#\$03	
	STAA	TR1	;10k Hz band
	LDD	#\$8404	;load up d
	STD	SPCR1	;turn on QSPI
	BGND		;go back to EVB16 software ;reassemble code for T125 to T10K ;experiment with different values
	BRA	T125	;branch back to TR125 line

Figure 16 QSPI_LED.ASM Flowchart

The Periodic Interrupt Timer (INT_TEST.ASM)

The periodic interrupt timer (PIT) is an internal timer that can be programmed to make an interrupt service request at specific intervals. One application of the PIT is to configure it to interrupt the processor every second so that an interrupt service routine can update a clock.

INT_TEST.ASM produces a square wave on the port F pins of the MC68HC16Z1. The square wave has a set frequency determined by the PIT timeout period. The program uses the level six autovector and the PIT times out at 15.6 ms. Port F is initialized for discrete output, then the code enters a wait loop until the programmed interval elapses. The interrupt service routine creates the square wave. **Figure 17** is a flowchart of **INT_TEST.ASM**.

For detailed information concerning interrupts, the PIT, and port F, refer to the *MC68HC16Z1 User 's Manual* (MC68HC16Z1UM/D), the *SIM Reference Manual* (SIMRM/AD), and the *CPU16 Reference Manual* (CPU16RM/AD).

INT_TEST.ASM Code Listing

	INCLUDE INCLUDE	'EQUATES.ASM' 'ORG00000.ASM'	;table of EQUates for common register addr ;initialize reset vector
	ORG	\$0200	;start program after interrupt vectors
* * * * *	Initiali	zation Routines	* * * *
	INCLUDE	'INITSYS.ASM'	;initially set EK=F, XK=O, YK=O, ZK=O ;set sys clock at 16.78 MHz, disable COP
	INCLUDE	'INITRAM.ASM'	;initialize and turn on SRAM ;set stack (SK=1, SP=03FE)
* * * * *	Initial	ize level 6 auto	vector address
	LDAB	#\$00	ek extension pointer - bank0
	LDD	#INT RT	iload Dacc with interrupt vector addr
	STD	\$002C	store addr to level 6 autovector
* * * * *	Initial	ize PortF	****
	LDAB	#\$0F	
	TBEK	#¢00	;ek extension pointer = banki
	STAB	#\$00 PFPAR #\$FF	;define port f as discrete i/o
	STAA	H O D R F	define port f as all output
	STAA	PORTF0	store \$ff to port f
* * * * *	Initial	ize the PIT	****
	LDD	#\$0616	
	STD	PICR	;pirql=6, piv=\$16
	עעם מדצ	#\$0000 DTTR	set the periodic timer at 15 6msec
	ANDP	#\$FF1F	;set interrupt priority to 000
* * * * *	Infinit	e loop	****
LOOP	NOP		;create an infinite loop
	BRA	LOOP	; waiting for interrupts
* * * * *	Excepti	ons/Interrupts	****
INT_RT	PSHM	D,CCR	;stack Dacc and CCR on stack
_	COM	PORTF0	;one's complement Port F, create square wave
	PULM	D,CCR	;pull Dacc and CCR from stack
	RTI		;return from interrupt

AN1233 F17

Figure 17 INT_TEST.ASM Flowchart

Signal Peak Detector (PEAK.ASM)

The signal peak detector graphically measures and displays the peak amplitude of a signal in real time. An audio signal is sampled at 24.95 kHz. The peak amplitude of the signal is detected, then a value that represents the peak on a bar of eight light-emitting diodes (LED) is generated. A reference value of 0.775 Vrms equivalent to 0 dB is used to relate the digital peak value to the LED display. The LED bar can display a signal in the range –15 dB to +6 dB, in 3 dB steps. **Figure 18** shows relationships between the LED bar, decibels, Vrms, and Vp. **Figure 19** shows the relationship between an analog input signal and the peak values displayed. **Figure 20** is a flowchart of **PEAK.ASM**.

$$dB = 20 \bullet \log\left(\frac{Vin}{Vref}\right)$$
$$0dB \ge Vref = 0.775 Vrms$$
$$Vneak = \sqrt{2} \bullet Vrms$$

AN1233 F18

Figure 18 Relationship Between Signal Amplitude and LED Bar

PEAK.ASM code must be downloaded to the EVB in a slightly different manner than usual. The code must be stored in the SRAM array, so it is important to enable and initialize SRAM module correctly before downloading. The steps listed below must be followed when downloading **PEAK.ASM**, **1K_FLTR.ASM**, and **5BAND_SA.ASM**.

- 1. Download the code
- 2. Set the IP to \$200
- 3. Trace the code until you have executed the section labeled 'RAM and Stack Initialization'
- 4. Set the IP to \$200
- 5. Download the program again.

The code originating in the internal RAM will now be correctly loaded into the MC68HC16Z1.

PEAK.ASM reads values from a look-up table in memory. The file **OUTVAL2.ASM** contains the table. Be sure this file is in the same directory as **PEAK.ASM** before assembly.

Figure 19 Analog Input vs Peak Display Level

After initializing the SRAM, the ADC, the QSPI, and the PIT, the code jumps to internal RAM at location \$F0000. Internal RAM access time is less than access time for the external RAM on the EVB. This extra speed is important to subsequent programs that use DSP routines.

The program then loops continuously, reading the ADC, encoding the ADC value to its equivalent LED value, and checking to see if the current value is greater than the previous peak value. If so, then the current peak value is updated and stored away in memory. The code can only increase the current peak value.

Peak value encoding is accomplished by self-modifying code that reads values from a look-up table in memory. Cycle counts for each instruction are given on the right-hand side comment line in **OUTVAL2.ASM**. They are used to determine the delay that is needed to create the 24.95-kHz sampling frequency.

The LED array is updated with the current peak value every 10.26 ms. This routine only detects and displays increases of the peak value. In order to follow a changing signal, the peak value must also be decreased periodically. A PIT interrupt performs this task every 62.5 ms. When the PIT times out, the interrupt service routine decrements the peak value and sends a new value to the display.

Using a PIT interrupt to decrement the peak value causes the LED display to decrease slowly, like a capacitor discharging, when the input signal decreases rapidly. This gives the display a more fluid appearance when rapidly-changing peak values are measured. If the display jumped from peak to peak, the discontinuity would lower the aesthetic appeal. In fact, most commercial audio analyzers show the relative peak differences of the frequency spectrum rather than attempt to display the peak signal precisely.

To test the code, hook up the system as shown in **Figure 11**. Input a known signal and observe the display. Apply an audio signal from the sound source and watch the peak detector execute in real time. If there is only one sound source output, connect it to either the left or right AFA input. The display is calibrated to the output of a CD player. The CD player puts out a line level signal, with .775 Vrms equal to 0 dB. If the sound source is not a CD player, adjust the output of the sound source so that the dynamic range of the signal is fully displayed.

PEAK.ASM Code Listing

	INCLUDE INCLUDE	'EQUATES.ASM' 'ORG00000.ASM'	<pre>;table of EQUates for common register addr ;initialize reset vector</pre>
**** PK CNT	Temporar EQU EQU	ry variable stora \$0200 \$0201	ge ; bank F ; bank F
	ORG	\$0200	
* * * * *	Initiali	zation Routines	****
	INCLUDE	'INITSYS.ASM'	;initially set EK=F, XK=0, YK=0, ZK=0 ;set sys clock at 16.78 MHz, disable COP
* * * * *	RAM and LDD	Stack Initializa #\$00FF	tion
	STD LDD	RAMBAH #\$0000	;store high ram array, bank F
	STD CLR LDAB	RAMBAL RAMMCR #\$0F	;store low ram array, 0000 ;enable ram
	TBSK LDS	#\$02FE	;set SK to bank F for system stack ;put SP in 1k internal SRAM
* * * * *	Initiali LDAB	lze level 6 autov #\$00	rector address
	TBEK LDD STD	#JMPINT \$002C	;ek extension pointer = bank0 ;load Dacc with interrupt vector addr ;store addr to level 6 autovector
* * * * *	Initiali	ze the PIT	****
	TBEK	#\$0F	;ek extension pointer = bankf
	STD	#\$0010 PICR #\$0101	;pirql=6, piv=\$16
	STD	HOLOI PITR HORIFIE	;set the periodic timer at 62.5msec
* * * * *	OCDI Ini	ticlication	*****
	LDAA STAA	#\$08 QPDR	;output pcs0/ss* to 0 when asserted
	STAA	UPAR	;assign QSM port pins to qspi module
	LDAA STAA	#ŞFE QDDR	;assign all QSM pins as outputs except miso
	LDD STD LDD STD	#\$8004 SPCR0 #\$0300 SPCR2	<pre>;mstr, womq=cpol=cpha=0 ;16 bits, 2.10MHz serial baud rate ;no interrupt generated, no wrap mode ;newqp=0, endqp=3, queued for 4 trans</pre>
* * * * *	Fill QSE	PI Command.ram to	write the config registers of the 14489
	STAA	CR0	<pre>;cont=1, bitse=1, pcs0=0, no delays needed</pre>

	STAA STAA LDAA	CR1 CR2 #\$40	
	STAA	CR3	;cont=0, bitse=1, pcs0=0, no delays needed
* * * * *	Fill QSI LDAA	PI Transmit.ram t #\$3F	to write the config registers of the 14489
	STD STD STD	TR0+1 TR2 TR3+1	;store \$3F to tran.ram registers
**** **** GO	Turn on of the M LDD	the QSPI, this w MC14489 drivers #\$8404	vill write to the config registers
SPIWT	STAA LDAA ANDA CMPA BNE	SPCR1 SPSR #\$80 #\$80 SPIWT	<pre>;turn on spi ;after sending data we wait until the ;spif bit is set, before we can send more ;check for spi done</pre>
* * * * *	Fill QSI	PI Command.ram to	o write the display registers of the 14489
	LDAA STAA	#\$C0 CR0	<pre>;cont=1, bitse=1, pcs0=0, no delays needed</pre>
	STAA LDAA STAA	CR1 #\$40 CR2	<pre>;cont=0, bitse=1, pcs0=0, no delays needed</pre>
	STAA LDAA STAA	CR4 #\$80 CR3	<pre>;cont=1, bitse=0, pcs0=0, no delays needed</pre>
* * * * *	Fill QSI The beg: LDD STD	PI Transmit.ram f inning LED values #\$8000 TPO	for display registers of the 14489 s will be \$00, all of the LEDs will be off :TRO - \$8000
	STAA LDD STD	TR3+1 #\$0080 TR1	<pre>;TR1 = \$0000 ;TR2 = \$0000 ;TR3 = \$XX80</pre>
	CLRD STD STD	TR2 TR4	;TR4 = \$0000
	LDD STD	#\$0400 SPCR2	;display registers need 5 transmissions ;newqp=0, endqp=4
* * * * *	ADC Init	tialization	****
	LDD STD LDD	#\$0000 ADCMCR #\$0003	; turn on ADC
	SID	ADCILO	,o-bit, set sample period
****	Initial: Set up t LDAB TBEK TBXK TBXK TBYK TBZK JMP	the extension reg #\$0F RAM	n registers for the internal ram in bank F gisters to point to bank F ;load b with \$0F ;transfer Bacc to Ek ;transfer Bacc to Xk ;transfer Bacc to Yk ;transfer Bacc to Zk ;jump to internal ram for speed!
* * * * *	Start of	E Internal 1K RAM	4
RAM	ORG CLR CLR	\$F0000 CNT PK	;clear LED update counter ;clear peak value
LP	CLRD STD	ADCTL1	<pre>; 2 clear Dacc ; 6 single 4 conversion, single channel AD0 ; writing to the ADCTL1 reg starts conv</pre>
	LDE	LJSRR0	; 6 load e with $x(n)$, left jus adc result0
*	Check if LDAA ADDA STAA BNE	E LEDs need updat CNT #1 CNT TRAN	<pre>ing ; 6 load Aacc with count ; 2 add 1 to Aacc ; 6 store new count ; 6,2 check to see if its time to update ; the LEDs, time = 256 * 668 cycles ; 668 cycles = 40.08usec ; so LED update time is 10.26msec</pre>

	LDD STD	#\$8404 SPCR1	; 6 load up d ; 6 turn on QSPI, send LED data out
* TRAN LD	Get LED TED STAA NOP NOP LDAA	encode value fro LD+3 LED_TBL	<pre>om look-up table ; 2 transfer Eacc to Dacc ; 6 Dacc high byte -> instruction ldaa \$03?? ; 2 no operation, wait for CPU pipeline ; 2 no operation, wait for CPU pipeline ; 6 load Aacc with the encoded LED value ; from scaled peak LED table</pre>
*	Update p CMPA BLS STAA STAA STAA STAA STAA STAA STAA	peak value if nee PK DN PK TR1 TR2 TR2+1 TR4 TR4+1	eded ; 6 compare value to previous peak value ; 6,2 branch if not more than peak value ; 6 store new peak value ; 6 store new value to all 5 qspi tran.rams ; 6 ; 6 ; 6
* * * * * * * * * * * * * * *	Loop to Clocks = N is the	generate calcula = 6 + 8*(N-1) N e number put into	ated delay >= 1 o the B accumulator
DN WAIT	LDAB DECB BNE	#\$4B WAIT	 75dec this loop will create an extra delay to make a 24.95kHz sampling rate or a 668 cycle sampling period 598 cycles
	JMP	LP	; 6 jump back to start another conversion
**** **** **** INT_RT	Exceptio This in represen PSHM LDAA BEQ	ons/Interrupts terrupt is used t nting the peak va D,CCR PK DONE	***** to decrement each LED bar value alue of the audio signal ;stack Dacc and CCR on stack ;load Aacc with peak value ;equal to 0?, then done
	ANDP RORA STAA STAA STAA STAA STAA STAA LDD STD	#\$FEFF TR1 TR2 TR2+1 TR4 TR4+1 PK #\$8404 SPCR1	<pre>;clear C bit ;rotate right once, decrease peak value ;store Aacc to all qspi tran.ram ;store Aacc to peak value ;load up Dacc ;turn on QSPI, send LED data out</pre>
DONE	PULM RTI	D,CCR	;pull Dacc and CCR from stack ;return from interrupt
**** JMPINT	Location ORG JMP	n of start of lev \$A000 INT_RT	vel 6 interrupt, has to be in bank 0
* * * * * * * * * * * * * * *	OUTVAL2 ADC read Encodes INCLUDE	is a 256 byte lo ling to a LED val to a scale of +6 'OUTVAL2.ASM'	ookup table to convert an lue that can be transmitted to the 14489 5, +3, 0, -3, -6, -9, -12, -15 dB ;LED Look up table

AN1233 F20A

Figure 20 PEAK.ASM Flowchart (Sheet 1 of 2)

Figure 20 PEAK.ASM Flowchart (Sheet 2 of 2)

A 1-kHz Bandpass Filter (1K_FLTR.ASM)

This code is similar in function to the peak detector, except that it executes a 1-kHz IIR bandpass filter on the input signal. The peak is detected and displayed on an LED bar in real time. The focus is on using the MC68HC16Z1 to implement the digital filter. **Figure 22** is a flowchart of **1K_FLTR.ASM**.

The objective is to take incoming sampled data x(n), and run the bandpass filter function on the sample to produce output y(n). Again, this is the basic 'black box' concept of electrical engineering — excite the input and watch the output change. The function in the 'black box' is defined below.

$$y(n) = 2 * \{ \alpha * [x(n) - x(n-2)] + \gamma * y(n-1) - \beta * y(n-2) \}$$

This function implements an IIR bandpass function with characteristics defined by the coefficients α , β , and γ . In an RLC bandpass filter circuit, resistors, capacitors, and inductors would characterize filter response. In the digital implementation of the filter, the α , β , and γ coefficients determine the response in much the same way.

The basic parameters that define digital filter response are the Q, the sampling frequency (Fs), and the center frequency (Fo). The Q value defines the sharpness of the filter and is equal to the center frequency divided by the bandwidth between the 3 dB points. The specified sampling frequency is 24.95 kHz, the center frequency is 1 kHz, and Q value is 1.5. **Figure 21** illustrates these relationships. **Table 3** shows the way in which coefficients are stored in memory.

XN1_1K	x(n – 1)	YN1_1K	y(n – 1)	GAM_1K	γ
XN2_1K	x(n – 2)	YN2_1K	y(n – 2)	BET_1K	$-\beta^1$
		X_2_1K	x(n) - x(n - 2)	ALP_1K	α

1. To speed processing, the calculated β coefficient is made negative, then added to the expression, rather than subtracted as shown in the equation above.

Equations that define the coefficients are shown below. Coefficient values are also given in the code listing.

$$\begin{split} \theta &= \{(2*\pi*Fo) \ / \ Fs \} \\ &X &= \theta \ / \ (2*Q) \\ &\text{If } X > \pi \ / \ 4 \ then \ X &= 0.75398 \\ &\beta &= 0.5*\{1 - tan \ (X)\} \ / \ \{1 + tan \ (X)\} \\ &\gamma &= (0.5 + \beta) * \cos \theta \\ &\alpha &= (0.5 - \beta) \ / \ 2 \end{split}$$

Where:

Fo = 1 kHz Fs = 24.95 kHz Q = 1.5

For more information concerning these equations, refer to Motorola Application Note *Digital Stereo 10-Band Generator* (APR2/D).

Once coefficient values have been obtained, they must be encoded. The assembler does not understand fractional decimal numbers, so fractional values are converted into signed 16-bit hexadecimal values. When using two's complement arithmetic, the most significant bit (bit 15) is the sign bit, and the fraction is contained in bits 14 to 0. Fifteen bits can represent the decimal numbers from 0 to 32,767. Multiply the decimal fraction by 32,768, then convert the value to the hexadecimal equivalent. Make certain that hexadecimal equivalents of negative values are in two's complement form. An example is given below.

Decimal fraction = 0.5

Multiply fractional decimal value by 32,768

0.5 * 32,768 = 16,384

Change decimal value to hexadecimal and binary values

16,384 dec = 4000 hex = 0100 0000 0000 0000 bin

4000 hex is the 16-bit fractional value.

CPU16 multiply and add instructions are used to implement the function. Processing is streamlined so that, in the final AFA design, five filters can be implemented in the 40.08 µs sampling period. For a more thorough discussion of the DSP instruction set and related CPU16 architecture, please consult Chapter 11 in the *CPU16 Reference Manual* (CPU16RM/AD). The processing sequence is as follows.

The ADC value x(n) is divided by two to prevent overflow.

The subtraction operation, x(n) - x(n - 2) is performed and the result is stored in location X_2_1K.

Three MAC instructions are executed, starting at address YN1_1K.

The value y(n-1) is multiplied by γ and added to the M accumulator.

The value y(n - 2) is multiplied by $-\beta$ and added to the M accumulator.

The value [x(n) - x(n-2)] is multiplied by α and added to the M accumulator.

The M accumulator is multiplied by two, using a left shift instruction, to obtain the y(n) value.

The x and y terms are updated before the next sample is processed:

x(n-1) becomes x(n-2) and x(n) becomes x(n-1)

y(n - 1) becomes y(n - 2) and y(n) becomes y(n - 1)

As mentioned earlier, the 1-kHz bandpass filter is very similar to the peak detector design. Once the DSP is finished on the input x(n) sample, the peak detect algorithm is executed.

The include file **OUTVAL1.ASM** is used to encode the DSP output with an LED display value multiplied by two. Be sure this file is in the same directory as 1K_FLTR.ASM during assembly.

The best way to test this program is to connect a signal generator with sine-wave sweep capability to the AFA inputs, then set it to sweep from 0 to 15 kHz. The 1-kHz LED bar should display the amplitude of a pure 1-kHz tone and the routine should filter out higher and lower frequency signals. Since Q is equal to 1.5, some side-lobe frequencies in the pass band should be evident. For instance, if a 2-kHz pure signal is sent into the filter, the side-lobe response of the 1-kHz bandpass will pass an attenuated level of the 2-kHz tone.

1K_FLTR.ASM Code Listing

	INCLUDE INCLUDE	'EQUATES.ASM' 'ORG00000.ASM'	;table of EQUates for common register addr ;initialize reset vector
***** COEFBS GAM_1K BETA_1K ALPH_1K	Addresse EQU EQU EQU EQU ORG dc.w dc.w dc.w	es of coefficient \$0280 COEFBS+\$0 COEFBS+\$2 COEFBS+\$4 \$F0280 \$7257 \$C9F0 \$04F7	s for the IIR Filters and initialization ;base addr of coefficients ;addr of the gamma coef ;addr of the beta coef ;addr of the alpha coef ;lk Hz gamma coef, Q=1.5 ;lk Hz beta coef, Q=1.5 ;lk Hz alpha coef, Q=1.5
***** XTRMBS XN1_1K XN2_1K	Addresse EQU EQU EQU ORG dc.w dc.w	es of filter term \$02A0 XTRMBS+\$0 XTRMBS+\$2 \$F02A0 \$0000 \$0000	<pre>ns for the x(n) terms and initialization ;base addr of x(n) filter terms ;x(n-1) ;x(n-2) ;lk Hz x(n-1) ;lk Hz x(n-2)</pre>
***** YTRMBS YN1_1K YN2_1K X_2_1K	Addresse EQU EQU EQU ORG dc.w dc.w dc.w	es of filter term \$02C0 YTRMBS+\$0 YTRMBS+\$2 YTRMBS+\$4 \$F02C0 \$0000 \$0000 \$0000	<pre>ns for the y(n) terms and initialization ;base addr of y(n) filter terms ;y(n-1) ;y(n-2) ;x(n) - x(n-2), stored here for mac ;1k y(n-1) ;1k y(n-2) ;1k [x(n) - x(n-2)]</pre>
***** PKRES PK_1K CNT AD	Addresse EQU EQU EQU EQU ORG dc.w dc.w	es of various tem \$02E0 PKRES+\$0 PKRES+\$1 PKRES+\$2 \$F02E0 \$0000 \$0000 \$0200	porary variables and initialization ;base addr of filter result storage ;peak value for 1k Hz ;count value for LED qspi update routine ;divided by two adc reading ;1k peak value, update count value ;divided by two adc location
****	Initializ	ation Routines	****
	INCLUDE	'INITSYS.ASM'	;initially set EK=F, XK=0, YK=0, ZK=0 ;set sys clock at 16.78 MHz, disable COP

* * * * *	RAM and	Stack Initializ	ation
	LDD	#\$00FF	
	STD	RAMBAH	;store high ram array, bank F
	LDD	#\$0000 Davidati	
	STD	RAMBAL	istore low ram array, 0000
	LDAR	H¢OF	, enable ram
	TRSK	#\$UI	set SK to bank E for system stack
	LDS	#\$02FE	;put SP in 1k internal SRAM
* * * * *	Initial	ize level 6 auto	vector address
	LDAB	#\$00	
	TBEK		;ek extension pointer = bank0
		#JMPINT	; load Dacc with interrupt vector addr
	SID	ŞUUZC	, store addr to level 6 autovector
* * * * *	Tnitial	ize the PIT	* * * * *
	LDAB	#\$0F	
	TBEK		;ek extension pointer = bankf
	LDD	#\$0616	
	STD	PICR	;pirql=6, piv=\$16
	LDD	#\$0101	
	STD	PITR	;set the periodic timer at 62.5msec
	ANDP	#\$F.F.TF.	;set interrupt priority to 000
* * * * *	OGDT TO	itialization	* * * *
		#\$08	
	STAA	OPDR	joutput pcs0/ss* to 0 when asserted
	LDAA	#\$OF	foucput peper bb to t when abberted
	STAA	QPAR	assign QSM port pins to gspi module;
	LDAA	#\$FE	
	STAA	QDDR	;assign all QSM pins as outputs except miso
		140004	
		#\$8004 CDCD0	imstr, womq=cpol=cpna=0
	SID	3PCR0 #\$0300	in interrupt generated no wrap mode
	עעם מידצ	#30300 SPCR2	inewon=0 endop=3 queued for 4 trans
	512	51 OILE	Hew The of end the of Angeles for I stand
* * * * *	Fill QS	PI Command.ram t	o write the config registers of the 14489
	LDAA	#\$C0	
	STAA	CR0	<pre>;cont=1, bitse=1, pcs0=0, no delays needed</pre>
	STAA	CR1	
	STAA	CR2	
		#\$40 CD2	isout 0 bitso 1 mas 0 0 mo dolour mooded
	SIAA	CRS	, cont=0, bitse=1, pcs0=0, no delays needed
* * * * *	Fill OS	PI Transmit.ram	to write the config registers of the 14489
	LDAA	#\$3F	
	STD	TR0+1	;store \$3F to tran.ram registers
	STD	TR2	
	STD	TR3+1	
	_		
*****	Turn on	the QSPI, this	will write to the config registers
CO		#c14489 drivers	
GO	STAA	#30404 9DCP1	turn on sni
SPIWT	LDAA	SPSR	after sending data we wait until the
01101	ANDA	#\$80	ispif bit is set, before we can send more
	CMPA	#\$80	icheck for spi done
	BNE	SPIWT	
* * * * *	Fill QS	PI Command.ram t	o write the display registers of the 14489
	LDAA	#\$C0	
	STAA	CR0	<pre>;cont=1, bitse=1, pcs0=0, no delays needed</pre>
	S'I'AA	CRT #040	
	LUAA CTN N	#\$4U CP2	,cont=0, bitse=1, pcs0=0, no delays needed
	SIAA	CR4	
	LDAA	#\$80	;cont=1, bitse=0, pcs0=0, no delays needed
	STAA	CR3	. cons 1, store o, post-o, no actays needed

****	Fill QSH The beg: LDD STD STAA LDD STD CLRD STD STD	PI Transmit.ram f inning LED values #\$8000 TR0 TR3+1 #\$0080 TR1 TR2 TR4	<pre>Eor display registers of the 14489 s will be \$00, all of the LEDs will be off ;TR0 = \$8000 ;TR1 = \$0080 ;TR2 = \$0000 ;TR3 = \$XX80 ;TR4 = \$0000</pre>
	LDD STD	#\$0400 SPCR2	;display registers need 5 transmissions ;newqp=0, endqp=4
* * * * *	ADC Init LDD STD LDD STD	tialization #\$0000 ADCMCR #\$0003 ADCTL0	***** ;turn on ADC ;8-bit, set sample period
****	Initial: Set up t LDAB TBEK TBXK TBXK TBYK TBZK JMP	ize the extension the extension reg #\$0F RAM	n registers for the internal ram in bank F gisters to point to bank F ;load b with \$0F ;transfer Bacc to Ek ;transfer Bacc to Xk ;transfer Bacc to Yk ;transfer Bacc to Zk ;jump to internal ram for speed!
**** RAM	Start of ORG CLR CLR	E Internal 1K RAM \$F0000 CNT PK_1K	/ ;clear LED update counter ;clear 1K peak value
*	Initial: ORP CLRD TDMSK LDY LDX	ization for DSP #\$0010 #COEFBS #YTRMBS	<pre>;set saturation mode for Macc ;clear Dacc ;no modulo addressing ;load y with the coef base addr ;load x with the yterm base addr</pre>
LP	CLRD STD	ADCTL1	; 2 clear Dacc ; 6 single 4 conversion, single channel ADO ; writing to the ADCTL1 reg starts conv
*	Divide : LDAA ASRA STAA	input x(n) by 2, LJSRRO AD	no overflow problem ; 6 load Aacc with left jus signed ADC value ; 2 divide by 2 ; 6 store divide by 2 adc value away
*	Check if LDAA ADDA STAA BNE LDD	E LEDs need updat CNT #1 CNT TRAN #\$8404	<pre>ing ; 6 load Aacc with count ; 2 add 1 to Aacc ; 6 store new count ; 6,2 check to see if its time to update ; the LEDs, time = 256 * 668 cycles ; 668 cycles = 40.08usec ; so LED update time is 10.26msec ; 6 load up d</pre>
TRAN	STD LDHJ	SPCR1	<pre>; 6 turn on QSPI, send LED data out ; 8 load h and i multiplier and multiplicand</pre>
F1K	CLRM LDE	AD	<pre>; 2 clear Macc ; 6 load Eacc with AD</pre>

*	Digital	processing algor	ithm
	SUBD	XN2_1K	; 2 transfer Eace to Dace ; 6 Dace = $x(n) - x(n-2)$
	STD LDD	X_2_1K XN1_1K	<pre>; 6 store Dacc to [x(n) - x(n-2)] addr ; 6 load Dacc with x(n-1)</pre>
	STED	XN1_1K	; 8 store $x(n)$ to $x(n-1)$ and ; store $x(n-1)$ to $x(n-2)$
	MAC	2.2	;12 gamma*(vn1)+Macc=Macc
	MAC	2,2	;12 beta* $(yn2)$ +Macc=Macc
	MAC TMET ASLE	-4,-4	<pre>; 2 transfer Macc to Eacc, truncate ; 2 multiply Eacc by 2</pre>
*	Get LED	encode value fro	m look-up table
	STAA	LD1K+3	<pre>; 2 transfer Eacc to Dacc ; 6 Dacc high byte -> instruction ldaa \$03?? ; 2 page state to CDU singling</pre>
	NOP		; 2 no operation, due to CPU pipeline ; 2 no operation, due to CPU pipeline
LD1K	LDAA	LED_TBL	; 6 load Aacc with the encoded LED value ; from scaled peak LED table
*	Update p	peak value if nee	eded
	BLS	DN1K	; 6,2 branch if not more than peak value
	STAA STAA	PK_1K TR2+1	; 6 store new peak value ; 6 store new value to 1k qspi tran.ram
*	Update y	y(n-1) and $y(n-2)$	
DNIK	LDD STED	YNI_IK YNI_1K	; 6 load Dacc with y(n-1) ; 8 store Eacc to y(n-1), Dacc to y(n-2)
* * * * *	Loop to	generate calcula	ated delay
* * * * *	Clocks = N is the	= 6 + 8*(N-1)	>= 1 o the B accumulator
መለ ተጥ	LDAB	#\$3D	; 61 this loop will create an extra delay
WALI	BNE	WAIT	; or a 668 cycle sampling period
	NOP		; 2
	NOP		; 2; 2; 2; 2; 2; 2; 2; 2; 2; 2; 2; 2; 2;
	JMP	LP	; 6 jump back to start another conversion
* * * * *	Exceptio	ons/Interrupts	****
* * * * *	represer	terrupt is used t nting the peak va	lue of the 1k filter band
INT_RT	PSHM LDAA	D,CCR PK 1K	;stack Dacc and CCR on stack ;load Aacc with 1K peak value
	BEQ	DONE	;equal to 0?, then done
	ANDP	#\$FEFF	;clear C bit
	STAA	TR2+1	store Aacc to 1k Hz qspi tran.ram
	STAA LDD	PK_1K #\$8404	;store Aacc to 1k Hz peak value ;load up Dacc
	STD	SPCR1	;turn on QSPI, send LED data out
DONE	PULM RTI	D,CCR	;pull Dacc and CCR from stack ;return from interrupt
* * * * *	Location	n of start of lev	rel 6 interrupt, has to be in bank 0
JMPINT	JMP	SAUUU INT_RT	
* * * * *	OUTVAL1	is a 256 byte lo	ookup table to convert an
* * * * *	Multipl:	ies by two and	ue that can be transmitted to the 14489
* * * * *	Encodes INCLUDE	to a scale of +6 'OUTVAL1.ASM'	5, +3, 0, -3, -6, -9, -12, -15 dB ;LED look up table

Figure 22 1K_FLTR.ASM Flowchart (Sheet 1 of 4)

AN1233 F22B

Figure 22 1K_FLTR.ASM Flowchart (Sheet 2 of 4)

AN1233 F22C

Figure 22 1K_FLTR.ASM Flowchart (Sheet 3 of 4)

AN1233 F20C

Figure 22 1K_FLTR.ASM Flowchart (Sheet 4 of 4)

The 5 Band Audio Frequency Analyzer (5BAND_SA.ASM)

The final design of the AFA is simple because of the groundwork that has already been done. **Figure 23** is a flowchart of **5BAND_SA.ASM**. Notice that five iterations of the IIR bandpass filter are executed before control passes to the interrupt routine.

The five bands and their Q values are: 125 Hz - 0.5, 500 Hz - 1.0, 1 kHz - 1.5, 4 kHz - 1.0, and 10 kHz - 0.5. Coefficient values are in the area labeled 'Address of coefficients...' at the beginning of the listing.

The specified Q values were chosen because they produce an appealing frequency display. If sharp filters with high Q values were used, the display would not show the relative differences between the bass, midrange, and treble frequency ranges. Energy associated to one particular frequency is not the primary concern of the AFA design, but rather the energy of an entire frequency band.

Test the code as before with the 1-kHz filter. Sweep a sinusoidal tone across the frequency and watch the appropriate LED array display signal energy. Apply a real time audio signal. Notice the differences between the high and low ends of the audio spectrum, the visible contrast between a bass drum and a cymbal.

5BAND_SA.ASM Code Listing

INCLUDE	'EQUATES.ASM'	;table of EQUates for common register addr
INCLUDE	'ORG00000.ASM'	; initialize reset vector
**** Address	es of coefficien	ts for the IIR Filters and initialization
COEFBS EQU	\$0280	;base addr of coefficients
GAM_125 EQU	COEFBS+\$0	;addr of the gamma coef
BET_125 EQU	COEFBS+\$2	;addr of the beta coef
ALP_125 EQU	COEFBS+\$4	;addr of the alpha coef
GAM_500 EQU	COEFBS+\$6	;addr of the gamma coef
BET_500 EQU	COEFBS+\$8	;addr of the beta coef
ALP_500 EQU	COEFBS+\$A	;addr of the alpha coef
GAM_1K EQU	COEFBS+\$C	;addr of the gamma coef
BET_1K EQU	COEFBS+\$E	;addr of the beta coef
ALP_1K EQU	COEFBS+\$10	;addr of the alpha coef
GAM_4K EQU	COEFBS+\$12	;addr of the gamma coef
BET_4K EQU	COEFBS+\$14	;addr of the beta coef
ALP_4K EQU	COEFBS+\$16	;addr of the alpha coef
GAM_10K EQU	COEFBS+\$18	;addr of the gamma coef
BET_10K EQU	COEFBS+\$1A	;addr of the beta coef
ALP_10K EQU	COEFBS+\$1C	;addr of the alpha coef
ORG	\$F0280	
dc.w	\$7C07	;125 Hz gamma coef, Q=0.5
dc.w	\$C3E9	;125 Hz beta coef, Q=0.5
dc.w	\$01F4	;125 Hz alpha coef, Q=0.5
dc.w	\$7774	;500 Hz gamma coef, Q=1.0
dc.w	\$C798	;500 Hz beta coef, Q=1.0
dc.w	\$03CB	;500 Hz alpha coef, Q=1.0
dc.w	\$7257	;lk Hz gamma coef, Q=1.5
dc.w	\$C9F0	;lk Hz beta coef, Q=1.5
dc.w	\$04F7	;lk Hz alpha coef, Q=1.5
dc.w	\$2C13	;4k Hz gamma coef, Q=1.0
dc.w	\$ED7A	;4k Hz beta coef, Q=1.0
dc.w	\$16BC	;4k Hz alpha coef,Q=1.0
dc.w	\$CA66	;10k Hz gamma coef, Q=0.5
dc.w	\$FDFE	;10k Hz beta coef, Q=0.5
dc.w	\$1EFE	;10k Hz alpha coef, Q=0.5

***** XTRMBS XN1 125	Addresse EQU EOU	es of filter tern \$02A0 XTRMBS+\$0	<pre>ns for the x(n) terms and initialization</pre>
XN2_125 XN1_500	EQU EQU	XTRMBS+\$2 XTRMBS+\$4	$i_{x(n-2)}$ $i_{x(n-1)}$
XN2_500	EQU	XTRMBS+\$6	$i \mathbf{x} (n-2)$
XN1_1K XN2_1K	EQU	XTRMBS+\$8 XTRMBS+\$A	ix(n-1) ix(n-2)
XN1_4K	EQU	XTRMBS+\$C	;x(n-1)
XN2_4K	EQU	XTRMBS+\$E	$i_{x(n-2)}$
XN1_10K XN2 10K	EQU	XTRMBS+\$10 XTRMBS+\$12	x(n-1) x(n-2)
	ORG	\$F02A0	
	ac.w dc.w	\$0000 \$0000	(125 Hz x(n-1)) (125 Hz x(n-2))
	dc.w	\$0000	;500 Hz x(n-1)
	dc.w dc.w	\$0000 \$0000	;500 Hz x(n-2) ;1k Hz x(n-1)
	dc.w	\$0000	;1k Hz x(n-2)
	dc.w	\$0000 \$0000	; 1k Hz $x(n-1)$
	dc.w	\$0000	(11 - 2) (11 Hz x(n-1))
	dc.w	\$0000	;1k Hz x(n-2)
* * * * *	Addresse	s of filter terr	ns for the y(n) terms and initialization
YTRMBS VN1 125	EQU	\$02C0 VTRMBS+\$0	;base addr of y(n) filter terms
YN2_125	EQU	YTRMBS+\$2	; y(n-2)
X_2_125	EQU	YTRMBS+\$4	x(n) - x(n-2), stored here for mac
YN2_500	EQU	YTRMBS+\$0	(y(n-2))
X_2_500	EQU	YTRMBS+\$A	ix(n) - x(n-2), stored here for mac
YN2_1K	EQU	YTRMBS+\$C	(y(n-2))
X_2_1K	EQU	YTRMBS+\$10	x(n) - x(n-2), stored here for mac
YNI_4K YN2 4K	EQU EQU	YTRMBS+\$12 YTRMBS+\$14	y(n-1) y(n-2)
X_2_4K	EQU	YTRMBS+\$16	x(n) - x(n-2), stored here for mac
YNI_IOK YN2 10K	EQU EQU	YTRMBS+\$18 YTRMBS+\$1A	;y(n-1) ;y(n-2)
X_2_10K	EQU	YTRMBS+\$1C	ix(n) - x(n-2), stored here for mac
	ORG dc.w	\$F02C0 \$0000	;125 Hz v(n-1)
	dc.w	\$0000	;125 Hz y(n-2)
	dc.w dc.w	\$0000 \$0000	;125 Hz [x(n) - x(n-2)] ;500 Hz v(n-1)
	dc.w	\$0000	;500 Hz y(n-2)
	dc.w dc.w	\$0000 \$0000	;500 Hz [$x(n) - x(n-2)$] ;1k Hz $y(n-1)$
	dc.w	\$0000	(111) $(112)(112)(112)(112)$
	dc.w	\$0000 \$0000	;1k Hz [$x(n) - x(n-2)$] :4k Hz $y(n-1)$
	dc.w	\$0000	(4k Hz y(n-2))
	dc.w	\$0000 \$0000	;4k Hz [x(n) - x(n-2)]
	dc.w	\$0000	;10k Hz y(n-2)
	dc.w	\$0000	;10k Hz [x(n) - x(n-2)]
* * * * *	Addresse	es of various ter	mporary variables and initialization
PKRES PK 125	EQU EOU	SUZEU PKRES+SO	;base addr of filter result storage ;peak value for 125 Hz
PK_500	EQU	PKRES+\$1	;peak value for 500 Hz
PK_1K dk 4k	EQU	PKRES+\$2 DKRES+\$3	;peak value for 1k Hz
PK_10K	EQU	PKRES+\$4	;peak value for 10k Hz
CNT	EQU	PKRES+\$6	; count value for LED qspi update routine
ΑU	ORG	\$F02E0	ATATAGE BY LWO ALL LEADING
	dc.w	\$0000 \$0000	;125 peak value, 500 peak value
	dc.w	\$0000 \$0000	;10k peak value
	dc.w	\$0000 \$0000	;update count value
	uc.w	\$UUUU	, urviued by two add reading
	ORG	\$0200	

* * * * *	Initial	ization Routines	* * * *
	INCLUDE	'INITSYS.ASM'	;initially set EK=F, XK=O, YK=O, ZK=O ;set sys clock at 16.78 MHz, disable COP
* * * * *	RAM and LDD	Stack Initializa #\$00FF	ation *
	STD LDD	RAMBAH #\$0000	;store high ram array, bank F
	STD CLR LDAB	RAMBAL RAMMCR #\$0F	;store low ram array, 0000 ;enable ram
	TBSK LDS	#\$02FE	;set SK to bank F for system stack ;put SP in 1k internal SRAM
* * * * *	Initial: LDAB	ize level 6 autov #\$00	vector address
	TBEK LDD STD	#JMPINT \$002C	;ek extension pointer = bank0 ;load Dacc with interrupt vector addr ;store addr to level 6 autovector
* * * * *	Initial: LDAB	ize the PIT #\$0F	****
	TBEK LDD	#\$0616	;ek extension pointer = banki
	STD LDD	PICR #\$0101	;pirql=6, piv=\$16
	STD ANDP	PITR #\$FF1F	;set the periodic timer at 62.5msec ;set interrupt priority to 000
* * * * *	QSPI In:	itialization	****
	STAA	QPDR #\$0F	;output pcs0/ss* to 0 when asserted
	STAA	QPAR térre	;assign QSM port pins to qspi module
	STAA	QDDR	;assign all QSM pins as outputs except miso
	LDD STD LDD STD	#\$8004 SPCR0 #\$0300 SPCR2	<pre>;mstr, womq=cpol=cpha=0 ;16 bits, 2.10MHz serial baud rate ;no interrupt generated, no wrap mode ;newqp=0, endqp=3, queued for 4 trans</pre>
* * * * *	Fill QS	PI Command.ram to	o write the config registers of the 14489
	STAA STAA STAA	#\$C0 CR0 CR1 CR2	<pre>;cont=1, bitse=1, pcs0=0, no delays needed</pre>
	LDAA STAA	#\$40 CR3	;cont=0, bitse=1, pcs0=0, no delays needed
* * * * *	Fill QSI	PI Transmit.ram t	to write the config registers of the 14489
	STD STD STD	#\$3F TR0+1 TR2 TR3+1	;store \$3F to tran.ram registers
* * * * * * * * * * GO	Turn on of the I LDD	the QSPI, this w MC14489 drivers #\$8404	will write to the config registers
SPIWT	STAA LDAA ANDA	SPCR1 SPSR #\$80	;turn on spi ;after sending data we wait until the ;spif bit is set, before we can send more
	CMPA BNE	#\$80 SPIWT	;check for spi done
* * * * *	Fill QSI LDAA	PI Command.ram to #\$C0	o write the display registers of the 14489
	STAA STAA	CR0 CR1	<pre>;cont=1, bitse=1, pcs0=0, no delays needed</pre>
	LDAA STAA	#\$40 CR2	<pre>;cont=0, bitse=1, pcs0=0, no delays needed</pre>
	STAA LDAA STAA	CR4 #\$80 CR3	<pre>;cont=1, bitse=0, pcs0=0, no delays needed</pre>

* * * * * * * * * *	Fill QSI The begin	PI Transmit.ram f inning LED values	for display registers of the 14489 s will be \$00, all of the LEDs will be off			
	STD STAA LDD STD CLPD	#\$8000 TR0 TR3+1 #\$0080 TR1	;TR0 = \$8000 ;TR1 = \$0080 ;TR2 = \$0000 ;TR3 = \$XX80 :TR4 = \$0000			
	STD STD	TR2 TR4	/IR4 - \$0000			
	LDD STD	#\$0400 SPCR2	;display registers need 5 transmissions ;newqp=0, endqp=4			
* * * * *	ADC Init	tialization #\$0000	****			
	STD	ADCMCR	;turn on ADC			
	LDD STD	#\$0003 ADCTL0	;8-bit, set sample period			
* * * * *	Initial	ize the extensior	registers for the internal ram in bank F			
*****	Set up 1	the extension reg #SOF	jisters to point to bank F :load b with SOF			
	TBEK	#\$01	itransfer Bacc to Ek			
	TBXK		;transfer Bacc to Xk			
	TBYK		;transfer Bacc to Yk			
	JMP	RAM	;jump to internal ram for speed!			
* * * * *	Start of Internal 1K RAM					
RAM	CLR	CNT	;clear LED update counter			
	CLR	PK_125	;clear 125 peak value			
	CLR	PK_500	;clear 500 peak value			
	CLR	PK_IK PK 4K	clear 4k peak value			
	CLR	PK_10K	;clear 10k peak value			
	CLRW	AD	;clear AD			
*	Initial	ization for DSP				
	CLRD	#\$0010	iset saturation mode for Macc			
	TDMSK		;no modulo addressing			
LP	LDY	#COEFBS	; 4 load y with the coef base addr			
	LDX	#YTRMBS	; 4 load x with the yterm base addr			
	CLRD		; 2 clear Dacc			
	STD	ADCTL1	; 6 single 4 conversion, single channel ADO			
			; writing to the ADCTL1 reg starts conv			
*	Divide :	input x(n) by 2, LDAA L	no overflow problem JSRR0 ; 6 load Aacc with left jus signed ADC value ; 2 divide by 2			
	STAA	AD	; 6 store divide by 2 adc value away			
*	Check i:	f LEDs need updat	ing			
	LDAA	CNT	; 6 load Aacc with count			
	ADDA Staa	#⊥ CNT	; 2 add 1 to Aacc ; 6 store new count			
	BNE	F125	; 6,2 check to see if its time to update			
			<pre>; the LEDs, time = 256 * 668 cycles ; 668 cycles = 40.08usec</pre>			
			; so LED update time is 10.26msec			
	LDD	#\$8404	; 6 load up Dacc			
	STD	SPCR1	; 6 turn on QSPI, send LED data out			
****	Start of	f the 125 Hz rout	ine			
гтүр	LDE	AD	; 6 load Eacc with AD			

*	Digital TED SUBD STD LDD STED	processing algor XN2_125 X_2_125 XN1_125 XN1_125 XN1_125	rithm ; 2 transfer Eacc to Dacc ; 6 Dacc = $x(n) - x(n-2)$; 6 store Dacc to $[x(n) - x(n-2)]$ addr ; 6 load Dacc with $x(n-1)$; 8 store $x(n)$ to $x(n-1)$ and ; store $x(n-1)$ to $x(n-2)$
	MAC MAC MAC TMER ASLE	2,2 2,2 2,2	<pre>;12 gamma*(yn1)+Macc=Macc ;12 beta*(yn2)+Macc=Macc ;12 alpha*[x(n)-x(n-2)]+Macc=Macc ; 6 transfer Macc to Eacc, round for converg ; 2 multiply Eacc by 2</pre>
* LD125	Get LED TED STAA NOP NOP LDAA	encode value fro LD125+3 LED_TBL	om look-up table ; 2 transfer Eacc to Dacc ; 6 Dacc high byte -> instruction ldaa \$03?? ; 2 no operation, due to CPU pipeline ; 2 no operation, due to CPU pipeline ; 6 load Aacc with the encoded LED value ; from scaled peak LED table
*	Update p CMPA BLS STAA STAA	peak value if nee PK_125 DN125 PK_125 TR4+1	eded ; 6 compare value to previous peak value ; 6,2 branch if not more than peak value ; 6 store new peak value ; 6 store new value to 125 qspi tran.ram
* DN125	Update y LDD STED	y(n-1) and y(n-2) YN1_125 YN1_125) ; 6 load Dacc with y(n-1) ; 8 store Eacc to y(n-1), Dacc to y(n-2)
***** F500	Start of CLRM LDE	E the 500 Hz DSP AD	routine ; 2 clear Macc ; 6 load Eacc with AD
*	Digital TED SUBD STD LDD STED MAC MAC MAC TMET ASLE	processing algor XN2_500 X_2_500 XN1_500 XN1_500 2,2 2,2 2,2 2,2	<pre>rithm ; 2 transfer Eacc to Dacc ; 6 Dacc = x(n) - x(n-2) ; 6 store Dacc to [x(n) - x(n-2)] addr ; 6 load Dacc with x(n-1) ; 8 store x(n) to x(n-1) and ; store x(n-1) to x(n-2) ;12 gamma*(yn1)+Macc=Macc ;12 beta*(yn2)+Macc=Macc ;12 alpha*[x(n)-x(n-2)]+Macc=Macc ; 2 transfer Macc to Eacc, truncate ; 2 multiply Eacc by 2</pre>
*	Get LED TED STAA NOP NOP	encode value fro LD500+3	om look-up table ; 2 transfer Eacc to Dacc ; 6 Dacc high byte -> instruction ldaa \$03?? ; 2 no operation, due to CPU pipeline ; 2 no operation, due to CPU pipeline
LD500	LDAA	LED_TBL	; 6 load Aacc with the encoded LED value ; from scaled peak LED table
*	Update p CMPA BLS STAA STAA	peak value if nee PK_500 DN500 PK_500 TR4	eded ; 6 compare value to previous peak value ; 6,2 branch if not more than peak value ; 6 store new peak value ; 6 store new value to 500 qspi tran.ram
* DN500	Update y LDD STED	y(n-1) and y(n-2) YN1_500 YN1_500) ; 6 load Dacc with y(n-1) ; 8 store Eacc to y(n-1), Dacc to y(n-2)
**** F1K	Start of CLRM LDE	f the 1k Hz routi AD	ine ; 2 clear Macc ; 6 load Eacc with AD

*	Digital TED SUBD STD LDD STED	processing algor XN2_1K X_2_1K XN1_1K XN1_1K XN1_1K	ithm ; 2 transfer Eacc to Dacc ; 6 Dacc = $x(n) - x(n-2)$; 6 store Dacc to $[x(n) - x(n-2)]$ addr ; 6 load Dacc with $x(n-1)$; 8 store $x(n)$ to $x(n-1)$ and ; store $x(n-1)$ to $x(n-2)$
	MAC MAC MAC TMET ASLE	2,2 2,2 2,2	<pre>;12 gamma*(yn1)+Macc=Macc ;12 beta*(yn2)+Macc=Macc ;12 alpha*[x(n)-x(n-2)]+Macc=Macc ; 2 transfer Macc to Eacc, truncate ; 2 multiply Eacc by 2</pre>
*	Get LED TED STAA NOP	encode value fro LD1K+3	m look-up table ; 2 transfer Eacc to Dacc ; 6 Dacc high byte -> instruction ldaa \$03?? ; 2 no operation, due to CPU pipeline
LD1K	NOP LDAA	LED_TBL	; 2 no operation, due to CPU pipeline ; 6 load Aacc with the encoded LED value ; from scaled peak LED table
*	Update j CMPA BLS STAA STAA	peak value if nee PK_1K DN1K PK_1K TR2+1	ded ; 6 compare value to previous peak value ; 6,2 branch if not more than peak value ; 6 store new peak value ; 6 store new value to 1k qspi tran.ram
* DN1K	Update y LDD STED	y(n-1) and y(n-2) YN1_1K YN1_1K	; 6 load Dacc with $y(n-1)$; 8 store Eacc to $y(n-1)$, Dacc to $y(n-2)$
**** F4K	Start o: CLRM LDE	f the 4k Hz routi AD	ne ; 2 clear Macc ; 6 load Eacc with AD
*	Digital TED SUBD STD LDD STED	processing algor XN2_4K X_2_4K XN1_4K XN1_4K XN1_4K	<pre>ithm ; 2 transfer Eacc to Dacc ; 6 Dacc = x(n) - x(n-2) ; 6 store Dacc to [x(n) - x(n-2)] addr ; 6 load Dacc with x(n-1) ; 8 store x(n) to x(n-1) and ; store x(n-1) to x(n-2)</pre>
	MAC MAC MAC TMET ASLE	2,2 2,2 2,2	<pre>;12 gamma*(yn1)+Macc=Macc ;12 beta*(yn2)+Macc=Macc ;12 alpha*[x(n)-x(n-2)]+Macc=Macc ; 2 transfer Macc to Eacc, truncate ; 2 multiply Eacc by 2</pre>
*	Get LED TED STAA NOP NOP	encode value fro LD4K+3	m look-up table ; 2 transfer Eacc to Dacc ; 6 Dacc high byte -> instruction ldaa \$03?? ; 2 no operation, due to CPU pipeline ; 2 no operation, due to CPU pipeline
LD4K	LDAA	LED_TBL	; 6 load Aacc with the encoded LED value ; from scaled peak LED table
*	Update j CMPA BLS STAA STAA	peak value if nee PK_4K DN4K PK_4K TR2	ded ; 6 compare value to previous peak value ; 6,2 branch if not more than peak value ; 6 store new peak value ; 6 store new value to 4k qspi tran.ram
* DN4K	Update y LDD STED	y(n-1) and y(n-2) YN1_4K YN1_4K	; 6 load Dacc with $y(n-1)$; 8 store Eacc to $y(n-1)$, Dacc to $y(n-2)$
***** F10K	Start o: CLRM LDE	f the 10k Hz rout AD	ine ; 2 clear Macc ; 6 load Eacc with AD

*	Digital	processing algor	it	hn	1
	TED		;	2	transfer Eacc to Dacc
	SUBD	XN2_10K	;	6	Dacc = x(n) - x(n-2)
	STD	X_2_10K	;	6	store Dacc to $[x(n) - x(n-2)]$ addr
	LDD	XN1_10K	;	6	load Dacc with x(n-1)
	STED	XN1_10K	;	8	store $x(n)$ to $x(n-1)$ and
			;		store $x(n-1)$ to $x(n-2)$
	MAC	2,2	;1	2	gamma*(yn1)+Macc=Macc
	MAC	2,2	;1	2	beta*(yn2)+Macc=Macc
	MAC	2,2	;1	2	alpha*[x(n)-x(n-2)]+Macc=Macc
	TMET		;	2	transfer Macc to Eacc, truncate
	ASLE		;	2	multiply Eacc by 2
*	Get LED	encode value fro	m	lc	ook-up table
	TED		;	2	transfer Eacc to Dacc
	STAA	LD10K+3	;	6	Dacc high byte -> instruction ldaa \$03??
	NOP		;	2	no operation, due to CPU pipeline
T D1 017	NOP			2	no operation, due to CPU pipeline
LDIOK	LDAA	TED_IRP	;	ю	from scaled peak LED table
			'		fiom scaled peak heb cabie
*	Update p	eak value			
	CMPA	PK_10K	;	6	compare value to previous peak value
	BLS	DN10K	;	6,	2 branch if not more than peak value
	STAA	РК_10К	;	6	store new peak value
	STAA	TR1	;	6	store new value to 10k qspi tran.ram
*	Update y	v(n-1) and $y(n-2)$			
DN10K	LDD	YN1_10K	;	6	load Dacc with y(n-1)
	STED	YN1_10K	;	8	store Eacc to $y(n-1)$, Dacc to $y(n-2)$
	NOP				
END	JMP	LP	;	6	jump back to start another conversion

* * * * * * * * * * * * * * *	Exception This interpresent	ons/Interrupts terrupt is used t nting the peak va	***** o decrement each LED bar value lue of each filter band	
INT_RT	PSHM	D,CCR	;stack Dacc and CCR on stack	
CK125	LDAA BEQ ANDP RORA STAA STAA	PK_125 CK500 #\$FEFF TR4+1 PK_125	<pre>;load Aacc with 125 peak value ;equal to 0?, then CK500 ;clear C bit ;rotate right once, decrease peak value ;store Aacc to 125 Hz qspi tran.ram ;store Aacc to 125 Hz peak value</pre>	
CK500	LDAA BEQ ANDP RORA STAA STAA	PK_500 CK1K #\$FEFF TR4 PK 500	<pre>;load Aacc with 500 peak value ;equal to 0?, then CK1K ;clear C bit ;rotate right once, decrease peak value ;store Aacc to 500 Hz qspi tran.ram ;store Aacc to 500 Hz peak value</pre>	
CK1K	LDAA BEQ ANDP RORA STAA STAA	PK_1K CK4K #\$FEFF TR2+1 PK_1K	<pre>;load Aacc with 1k peak value ;equal to 0?, then CK4K ;clear C bit ;rotate right once, decrease peak value ;store Aacc to 1k Hz qspi tran.ram ;store Aacc to 1k Hz peak value</pre>	
CK4K	LDAA BEQ ANDP RORA STAA STAA	PK_4K CK10K #\$FEFF TR2 PK_4K	<pre>;load Aacc with 4k peak value ;equal to 0?, then CK10K ;clear C bit ;rotate right once, decrease peak value ;store Aacc to 4k Hz qspi tran.ram ;store Aacc to 4k Hz peak value</pre>	
CK10K	LDAA BEQ ANDP RORA STAA STAA	PK_10K UPDATE #\$FEFF TR1 PK_10K	<pre>;load Aacc with 10k peak value ;equal to 0?, then UPDATE ;clear C bit ;rotate right once, decrease peak value ;store Aacc to 10k Hz qspi tran.ram ;store Aacc to 10k Hz peak value</pre>	
UPDATE	LDD STD	#\$8404 SPCR1	;load up Dacc ;turn on QSPI, send LED data out	
DONE	PULM RTI	D,CCR	;pull Dacc and CCR from stack ;return from interrupt	
**** JMPINT	Location ORG JMP	n of start of lev \$A000 INT_RT	rel 6 interrupt, has to be in bank 0	
* * * * * * * * * * * * * * * * * * * *	OUTVAL1 is a 256 byte lookup table to convert an ADC reading to a LED value that can be transmitted to the 14489 Multiplies by two and Encodes to a scale of +6, +3, 0, -3, -6, -9, -12, -15 dB			

INCLUDE 'OUTVAL1.ASM' ;LED Look up table

Figure 23 5BAND_SA.ASM Flowchart (Sheet 1 of 4)

AN1233 F22B

Figure 23 5BAND_SA.ASM Flowchart (Sheet 2 of 4)

AN1233 F22C

Figure 23 5BAND_SA.ASM Flowchart (Sheet 3 of 4)

CONCLUSION

This application note is intended to give designers some insight concerning the use of digital signal processing algorithms with a microcontroller. The finished project is flexible enough to permit experimenting with different filters and LED output displays. DSP allows the experimenter to make on-the-fly changes in filter response by changing the coefficients.

REFERENCES

The following Motorola documents are referred to in this application note.

- M68HC16Z1EVB User's Manual (M68HC16Z1EVB/D)
- MC68HC16Z1 User's Manual (MC68HC16Z1UM/D)
- CPU16 Reference Manual (CPU16RM/AD)
- QSM Reference Manual (QSMRM/AD)
- ADC Reference Manual (ADCRM/AD)
- MC14489 Data Sheet (MC14489/D)

These items can be obtained through a Motorola Sales Office or Literature Distribution Center.

Motorola reserves the right to make changes without further notice to any products herein. Motorola makes no warranty, representation or guarantee regarding the suitability of its products for any particular purpose, nor does Motorola assume any liability arising out of the application or use of any product or circuit, and specifically disclaims any and all liability, including without limitation consequential or incidental damages. "Typical" parameters can and do vary in different applications. All operating parameters, including "Typicals" must be validated for each customer application by customer's technical experts. Motorola does not convey any license under its patent rights nor the rights of others. Motorola products are not designed, intended, or authorized for use as components in systems intended for surgical implant into the body, or other applications intended to support or sustain life, or for any other application in which the failure of the Motorola product could create a situation where personal injury or death may occur. Should Buyer purchase or use Motorola products for any such unintended or unauthorized application, Buyer shall indemnify and hold Motorola and its officers, employees, subsidiaries, affiliates, and distributors harmless against all claims, costs, damages, and expenses, and reasonable attorney fees arising out of, directly or indirectly, any claim of personal injury or death associated with such unintended or unauthorized use, even if such claim alleges that Motorola was negligent regarding the design or manufacture of the part. *Motorola*, Inc. is an Equal Opportunity/Affirmative Action Employer.

TO OBTAIN ADDITIONAL PRODUCT INFORMATION:

USA/EUROPE:	Motorola Literature Distribution;
JAPAN:	Nippon Motorola Ltd.; Tatsumi-SPD-JLDC, Toshikatsu Otsuki,
HONG KONG:	6F Selbu-Butsuryu-Center, 3-14-2 Tatsumi Koto-Ku, Tokyo 135, Japan. 03-3521-8315 Motorola Semiconductors H.K. Ltd.; 8B Tai Ping Industrial Park, 51 Ting Kok Road, Tai Po, N.T. Hong Kong, 852-26629298
MFAX: INTERNET:	RMFAX0@email.sps.mot.com - TOUCHTONE (602) 244-6609 http://www.mot.com

